
1 Introduction

This guide is the third volume in the “Maxima via Examples” collection. Its purpose is to show how Maxima
can be used to solve differential equations, either symbolically or by means of numerical methods.
This document, together with the PDF and wxMaxima versions of this Guide to Differential Equations with
Maxima, is available on the https://maxima-french-doc.fr website.

In addition to Maxima’s standard commands for solving differential equations, Maxima also provides packages
containing solution methods for particular types of differential equations. Some of these packages will be
presented in this document.

This guide is based on Maxima version 5.49 and therefore also applies to all later versions. I would like to
thank Dr Wolfgang Lindner, whose book Introductory Differential Equations with Maxima was a valuable
source of inspiration in the preparation of this guide. I should also mention the manuals by Edwin L. Woollett,
entitled Maxima by Example, in particular Chapter 3, Ordinary Differential Equation Tools. Finally, the manual
wxMaxima for calculus volume 2, and especially Chapter 4, contains very useful information on this topic.

Throughout this guide, it is assumed that the functions under consideration are differentiable as many times as
necessary on the intervals under study. The chapters of this booklet deal with ordinary differential equations
(ODEs), characterized by the fact that the unknown function depends on only one variable, often denoted by x
or t. A differential equation relates this function to its successive derivatives with respect to that single variable.
The other category of differential equations concerns functions of several variables, whose equations involve
partial derivatives; these will be referred to as PDEs (partial differential equations).

If you have an Internet connection, the red links entitled Click here for Live Demo (just after Result :) loads
the Maxima commands presented on the Maxima server https://net124.reltub.ca/yamwi/ and enables you to
run them online. This feature allows you to experiment and modify parameters or variables without having to
install Maxima on your computer.

Eduardo Ochs, whom I thank for his collaboration, offers a live version of this guide thanks to the eev project.
In more concrete terms, eev is a library for Emacs that lets us create executable logs of what we do in a format
that is reasonably easy to read and to modify, and that lets us "play back" those logs step by step in any order.
This live eev version is accessible at: https://anggtwu.net/MAXIMA/2026-gosse.mac.html

2 Defining an ODE with Maxima

If y is a function of the variable x, its successive derivatives
dy

dx
,
d2y

dx2
, . . . are defined in Maxima respectively

by diff(y,x) and diff(y,x,2), . . .
These derivatives can therefore be used to write a differential equation, provided that diff is preceded by an
apostrophe ’ in order to prevent Maxima from computing the derivative immediately.

’diff(y,x) = -y;
’diff(y(x),x) = -y(x);
’diff(y,x) + (1/(2*x))*y = 2;
’diff(y,x,2) + y = 0;
’diff(f,x,2) + 3*’diff(f,x) = k*f;
Comment:

• The function in the differential equation can be denoted by y or y(x). However, the notation y(x)
is essential when several variables are involved.

• An ODE can be assigned to a variable with the command: eq:’diff(y,x)=-y.

Maxima command:

Result: Click here for Live Demo

(% i1) ’diff(y,x) = -y;

d

dx
y=− y (% o1)

(% i2) ’diff(y(x),x) = -y(x);

d

dx
y(x)=− y(x) (% o2)

(% i3) ’diff(y,x) + (1/(2*x))*y = 2;

d

dx
y+

y

2x
=2 (% o3)

(% i6) ’diff(y,x,2) + y = 0;

d2

dx2
y+ y=0 (% o6)

(% i7) ’diff(f,x,2) + 3*’diff(f,x) = k*f;

d2

dx2
f +3

(
d

dx
f

)
= fk (% o7)

3 Checking whether a function is a solution of an ODE

Principle: replace the unknown function in the differential equation by the function to be tested.

eq: (1+x^2)*’diff(y,x) + 2*x*y = 4*x^3
sol: y = (x^4+1)/(1+x^2);
verif: ev(eq, sol, diff);
ratsimp(verif);
Comment: The first command defines the ODE and the second a candidate solution. The third command
replaces y in the ODE by the expression defined in sol, then computes the derivatives. The final command
simplifies the result in order to determine whether the differential equation is satisfied.

Maxima command:

Result: Click here for Live Demo

(% i6) eq: (1+xˆ 2)*’diff(y,x) + 2*x*y = 4*xˆ 3;(
x2+1

) ( d

dx
y

)
+2xy=4x3 (eq)

(% i7) sol: y = (xˆ 4+1)/(1+xˆ 2);

y=
x4+1

x2+1
(sol)

(% i8) verif: ev(eq, sol, diff);

(
x2+1

) ( 4x3

x2+1
−

2x
(
x4+1

)
(x2+1)2

)
+

2x
(
x4+1

)
x2+1

=4x3 (verif)

(% i9) ratsimp(verif);

4x3=4x3 (% o9)

f(x):= (x^4+1)/(1+x^2);
(1+x^2)*diff(f(x),x)+2*x*f(x);
ratsimp(%);
Comment: A standard method is used in this example. We define the function f whose status as a
solution is to be checked, then explicitly compute the left-hand side of the ODE by replacing y with f(x)
and evaluating the derivative. It only remains to simplify the result in order to draw the conclusion.

Maxima command:

Result: Click here for Live Demo

(% i10) f(x):= (xˆ 4+1)/(1+xˆ 2);

f(x) :=
x4+1

1+x2
(% o10)

(% i11) (1+xˆ 2)*diff(f(x),x)+2*x*f(x);(
x2+1

) ( 4x3

x2+1
−

2x
(
x4+1

)
(x2+1)2

)
+

2x
(
x4+1

)
x2+1

(% o11)

(% i12) ratsimp(%);

4x3 (% o12)

test(fun, equation, var) := block(
[r],
r : subst(fun, y, equation),
r : ev(r, diff),
is(fullratsimp(lhs(r)-rhs(r)) = 0)

)$
eq2 : ’diff(y,x,2) + y = 0;
f(x):=k*sin(x);
test(f(x),eq2,x);
test(x*sin(x),eq2,x);
Comment: To automate the preceding calculation, we define a function that tests whether a function
is a solution of an ODE. Its parameters are the function to be tested, the differential equation, and the
differentiation variable.
For the calculation, we define a new differential equation eq2, a function f defined by f(x) = k sin(x),
and apply the test procedure. We do the same with the function x 7−→ x sin(x).

Maxima command:

Result: Click here for Live Demo

(% i18) eq2 : ’diff(y,x,2) + y = 0;

d2

dx2
y+ y=0 (eq2)

(% i19) f(x):=k*sin(x);

f(x) := k sin (x) (% o19)

(% i20) test(f(x),eq2,x);

true (% o20)

(% i21) test(x*sin(x),eq2,x);

false (% o21)

4 Solving an ODE with ode2, Maxima’s built-in solver

4.1 Examples of ode2 usage

Without considering special cases, theory shows that an ordinary differential equation has a family of solution
functions, called the general solution. For a first-order ODE (an equation defined by the first derivative of y),
the family of functions is determined up to one constant. For a second-order equation, it is determined up to
two constants. By default, Maxima denotes these constants by %c, %k1, . . .
The built-in Maxima command for finding the general solution of a first- or second-order ODE is
ode2(differential equation,unknown function,variable).

eq1 : ’diff(y,x) + 2*y = 6*x;
sol1 : ode2(eq1, y, x);
Comment: Direct application of the solving command for a first-order ODE.

Maxima command:

Result: Click here for Live Demo

(% i2) eq1 : ’diff(y,x) + 2*y = 6*x;
sol1 : ode2(eq1, y, x);

d

dx
y+2y=6x (eq1)

y=%e−(2x)

(
3%e2x (2x− 1)

2
+%c

)
(sol1)

Let us plot several curves from this family of solutions:

s1 : rhs(sol1)$
wxdraw2d(

xrange = [-2, 4],
yrange = [-5, 12],
xlabel = "x",
ylabel = "y",
title = "Solutions de l EDO pour différentes valeurs de %c",
color = red,
key = "%c = 0",
explicit(subst(0, %c, s1), x, -3, 3),
key = "%c = 1",
color = blue,
explicit(subst(1, %c, s1), x, -3, 3),
key = "%c = 2",

color = green,
explicit(subst(2, %c, s1), x, -3, 3),
key = "%c = 3",
color = orange,
explicit(subst(3, %c, s1), x, -3, 3),
key = "%c = 4",

color = violet,
explicit(subst(4, %c, s1), x, -3, 3)

)$

Maxima command:

Comment: The command rhs(sol1) extracts the right-hand side of the general expression for the
solution sol1, defined by y = ....

Result: Click here for Live Demo

eq2 : ’diff(y,x,2) - 3*’diff(y,x) + 2*y = 0;
sol2 : ode2(eq2, y, x);
Comment: Solving a second-order ODE.

Maxima command:

Result: Click here for Live Demo

(% i4) eq2 : ’diff(y,x,2) - 3*’diff(y,x) + 2*y = 0;
sol2 : ode2(eq2, y, x);

d2

dx2
y− 3

(
d

dx
y

)
+2y=0 (eq2)

y=%ex%k2+%e2x%k1 (sol2)

eq3 : ’diff(y,x,2) + y = sin(x);
sol3 : ode2(eq3, y, x);
Comment: A second-order ODE with a nonzero right-hand side.

Maxima command:

Result: Click here for Live Demo

(% i6) eq3 : ’diff(y,x,2) + y = sin(x);
sol3 : ode2(eq3, y, x);

d2

dx2
y+ y=sin (x) (eq3)

y=%k1 sin (x)− x cos (x)

2
+%k2 cos (x) (sol3)

4.2 The "method" variable of ode2

The method variable is a system variable in Maxima that is automatically set by ode2 (and contrib_ode)
after solving a differential equation. It contains the name of the method that was successfully used to solve
the equation. This information helps to understand how Maxima solved the differential equation, thus helping
students make the connection with the course material and the theory.

• For first-order differential equations, the method used can be linear, separable, exact, homogeneous,
Bernoulli...

• For second-order differential equations, the method used can be constant coefficients, exact, variation of
parameters, Euler...

eq1:’diff(y,x) + 2*y = 4*x;
ode2(eq1, y, x);
print("Method used: ", method)$
Comment: Example of using the linear method.

Maxima command:

Result: Click here for Live Demo

(% i20) eq1:’diff(y,x) + 2*y = 4*x;
ode2(eq1, y, x);
print("Method used : ", method)$

d

dx
y+2y=4x (eq1)

y=%e−(2x)
(
%e2x (2x− 1)+%c

)
(% o19)

"Method used : "
linear

eq2:’diff(y,x) + y = y^3;
ode2(eq2, y, x);
print("Method used : ", method)$
Comment: Example of using the separable variables method.

Maxima command:

Result: Click here for Live Demo

(% i9) eq2:’diff(y,x) + y = yˆ 3;
ode2(eq2, y, x);
print("Method used : ", method)$

d

dx
y+ y= y3 (eq2)

log (y+1)− 2 log (y)+ log (y− 1)

2
=x+%c (% o8)

"Method used : "
separable

5 Particular solution of a first-order ODE

If an additional condition is given (the value of the function at a point, the value of its derivative at a point, . . . ),
it is then possible to determine a unique function in the family of solutions of the first-order ODE that satisfies
this condition. This function is the particular solution satisfying the given condition.

Principle. First, solve the differential equation in order to obtain its general solution, which contains one
integration constant. Then apply the initial condition to this general solution in order to determine the value of
this constant. This gives the particular solution satisfying the given initial condition.

5.1 Computing a particular solution with Maxima

Find the particular solution fp of the ODE
dy

dx
+ 2y = x satisfying y(0) = 1. (In this case, we say that we are

solving an initial value problem.)

eq : ’diff(y,x) + 2*y = x$
sol:ode2(eq,y,x);
define(f(x),rhs(sol));
const:solve(f(0)=1,%c);
define(fp(x),subst(rhs(const[1]),%c,f(x)));
expand(fp(x));
Comment: We define the ODE, solve it with ode2, and define the general solution f . The constant %c is
found by solving an equation that takes the initial condition into account; then the particular solution fp
is defined.

Maxima command:

Result: Click here for Live Demo

(% i2) eq : ’diff(y,x) + 2*y = x;
sol:ode2(eq,y,x);

d

dx
y+2y=x (eq)

y=%e−(2x)

(
%e2x (2x− 1)

4
+%c

)
(sol)

(% i3) define(f(x),rhs(sol));

f(x) :=%e−(2x)

(
%e2x (2x− 1)

4
+%c

)
(% o3)

(% i4) const:solve(f(0)=1,%c); [
%c=

5

4

]
(const)

(% i5) define(fp(x),subst(rhs(const[1]),%c,f(x)));

fp(x) :=%e−(2x)

(
%e2x (2x− 1)

4
+

5

4

)
(% o5)

(% i6) expand(fp(x));

x

2
+

5%e−(2x)

4
− 1

4
(% o6)

Let us represent this initial value problem graphically. To do so, let us plot several solution functions and the
particular solution passing through the point with coordinates (0, 1).

wxdraw2d(
xrange = [-2, 4],
yrange = [-1, 12],
xlabel = "x",
ylabel = "y",
title = "Problème de Cauchy dy/dx+2y=x et y(0)=1",
color = red,
key = "%c =5/4",
explicit(subst(5/4, %c, f(x)), x, -3, 3),
color = blue,
key = "%c = 1",
explicit(subst(1, %c, f(x)), x, -3, 3),
color = green,
key = "%c = 2",
explicit(subst(2, %c, f(x)), x, -3, 3),
color = orange,
key = "%c = 3",
explicit(subst(3, %c, f(x)), x, -3, 3),
color = violet,
key = "%c = 4",

explicit(subst(4, %c, f(x)), x, -3, 3),
point_size=1,point_type=7,key="point (0,1) ",

color=black,points([[0,1]])
)$

Maxima Command :

Résultat : Click here for Live Demo

5.2 Using Maxima’s built-in ic1 command

In order for Maxima to compute directly the expression of the solution of the first-order differential equation
satisfying y(x0) = y0, the general solution must first be found with ode2. Then the command ic1 is used
with the following syntax:
ic1(general_solution, x=x0, y=y0)

eq2 : ’diff(y,x) - 3*y = 6*exp(x);y(0)=2;
solgen : ode2(eq2,y,x);
solpart : ic1(solgen,x=0,y=2);

Comment: Finding the particular solution of
dy

dx
− 3y = 6ex satisfying y(0) = 2.

Maxima command:

Result: Click here for Live Demo

(% i29) eq2 : ’diff(y,x) - 3*y = 6*exp(x);y(0)=2;

d

dx
y− 3y=6%ex (eq2)

y(0)= 2 (% o29)

(% i34) solgen : ode2(eq2,y,x);

y=%e3x
(
%c− 3%e−(2x)

)
(solgen)

(% i35) solpart : ic1(solgen,x=0,y=2);

y=5%e3x− 3%ex (solpart)

6 Particular solution of a second-order ODE

The general solution of a second-order ODE is defined by two parameters. If two conditions are imposed
(values of the function at certain points or values of its derivative at certain points, . . . ), it is then possible
to determine a unique function in the family of solutions of the second-order ODE that satisfies these two
conditions.

Principle. First, solve the differential equation with ode2 in order to obtain its general solution, which contains
two integration constants. Then apply the initial conditions to this general solution in order to determine the
values of these constants. This gives the particular solution satisfying the initial value problem for this type of
ODE.

6.1 Computing a particular solution with Maxima

This method applies in all cases where the particular function is defined by two conditions, unlike the built-in
commands, which apply only in certain cases.

Find the particular solution of
d2y

dx2
+ y = 0 with y(0) = 2 and y′

(
π
4

)
= 1

eq: ’diff(y,x,2)+y = 0;y(0)=2;yprime(%pi/4)=1;
sol_gen: ode2(eq, y, x);
define(f(x),rhs(sol_gen));
define(fprime(x),diff(f(x),x));
constantes:solve([f(0)=2,fprime(%pi/4)=1],[%k1,%k2]);
subst(constantes,f(x));
Comment: Defining f ′ as a function subsequently simplifies the solution of the system of equations used
to determine the integration constants.

Maxima command:

Result: Click here for Live Demo

(% i20) eq: ’diff(y, x, 2) + y = 0;y(0)=2;yprime(%pi/4)=1;
sol_gen: ode2(eq, y, x);

d2

dx2
y+ y=0 (eq)

y(0)= 2 (% o18)

yprime
(π
4

)
=1 (% o19)

y=%k1 sin (x)+%k2 cos (x) (sol_ gen)

(% i5) define(f(x),rhs(sol_gen));

f(x) :=%k1 sin (x)+%k2 cos (x) (% o5)

(% i6) define(fprime(x),diff(f(x),x));

fprime(x) :=%k1 cos (x)−%k2 sin (x) (% o6)

(% i9) constantes:solve([f(0)=2,fprime(%pi/4)=1],[%k1,%k2]);[[
%k1=

√
2+2 ,%k2=2

]]
(constantes)

(% i16) subst(constantes,f(x)); (√
2+2

)
sin (x)+ 2 cos (x) (% o16)

6.2 Using Maxima’s built-in ic2 command

The command ic2 determines the particular solution when the two conditions give the values y(x0) = y0 and
y′(x0) = yp0, both known at the same point x0. The command syntax is:
ic2(general_solution, x=x0, y=y0, ’diff(y,x)=yp0)

Find the particular solution of
d2y

dx2
+ 3

dy

dx
+ 2y = 0 with y(0) = 1 and y′(0) = 0. In this case, there are two

initial conditions.

eq: ’diff(y,x,2) + 3*’diff(y,x) + 2*y = 0;
sol_gen: ode2(eq,y,x);
sol_part: ic2(sol_gen, x=0, y=1, ’diff(y,x)=0);

Maxima command:

Result: Click here for Live Demo

(% i16) eq: ’diff(y, x, 2) + 3*’diff(y, x) + 2*y = 0;y(0)=1;yprime(0)=0;
sol_gen: ode2(eq, y, x);
sol_part: ic2(sol_gen, x=0, y=1, ’diff(y,x)=0);

d2

dx2
y+3

(
d

dx
y

)
+2y=0 (eq)

y(0)= 1 (% o13)

yprime(0)= 0 (% o14)

y=%e−(2x)%k2+%e−x%k1 (sol_ gen)

y=2%e−x−%e−(2x) (sol_ part)

6.3 Using Maxima’s built-in bc2 command

The command bc2 solves a boundary value problem. In this case, the two conditions used to find the particular
solution are given by the values of y at two points x1 and x2: y(x1) = y1 and y(x2) = y2. The command
syntax is:
bc2(solution, x=x1, y=y1, x=x2, y=y2)

Find the particular solution of x2
d2y

dx2
− 2x

dy

dx
+ 2y = x3 with y(1) = 1 and y(2) = 4.

eq: x^2*’diff(y,x,2) - 2*x*’diff(y,x) + 2*y = x^3;
sol_gen: ode2(eq,y,x);
sol_bord: bc2(sol_gen, x=1, y=1, x=2, y=4);

Maxima command:

Result: Click here for Live Demo

(% i29) eq: xˆ 2*’diff(y, x, 2) - 2*x*’diff(y, x) + 2*y = xˆ 3;y(1)=1;y(2)=4;
sol_gen: ode2(eq, y, x);
sol_bord: bc2(sol_gen, x=1, y=1, x=2, y=4);

x2
(

d2

dx2
y

)
− 2x

(
d

dx
y

)
+2y=x3 (eq)

y(1)= 1 (% o26)

y(2)= 4 (% o27)

y=
x3

2
+%k1x2+%k2x (sol_ gen)

y=
x3

2
− x2

2
+x (sol_ bord)

7 Study of some types of differential equations

7.1 Directly integrable differential equation: y′ = f(x)

Solve the differential equation y′ = x2 + cos(x)

f(x):=x^2+cos(x);
eq: ’diff(y, x) = f(x);
sol_gen: ode2(eq,y,x);
y=integrate(f(x),x)+%c;
Comment: Define the function f and solve the equation with ode2. Another method consists of inte-
grating the function f directly, without forgetting to add the integration constant.

Maxima command:

Result: Click here for Live Demo

(% i1) f(x):=xˆ 2+cos(x);

f(x) :=x2+cos (x) (% o1)

(% i3) eq: ’diff(y, x) = f(x);
sol_gen: ode2(eq, y, x);

d

dx
y=cos (x)+x2 (eq)

y=sin (x)+
x3

3
+%c (sol_ gen)

(% i4) y=integrate(f(x),x)+%c;

y=sin (x)+
x3

3
+%c (% o4)

7.2 Autonomous differential equation y′ = f(y)

In this case, the variable x does not appear explicitly in the equation. The equation is usually solved with ode2;
then y is isolated by solving the resulting equation for y. It is often necessary to transform the result in order to
obtain a simplified expression.

Solve the differential equation y′ = y(1− y)

eq : ’diff(y,x) = y*(1-y);
sol : ode2(eq, y, x);
sol_simp : logcontract(sol);
sol_exp : exp(lhs(sol_simp)) = exp(rhs(sol_simp));
sol_y : solve(sol_exp, y);
Comment: Maxima must be guided through the required transformations. To do this, apply the exponen-
tial manually to both sides of the obtained solution, then solve using the solve command.

Maxima command:

Result: Click here for Live Demo

(% i44) eq : ’diff(y,x) = y*(1-y);
sol : ode2(eq, y, x);
sol_simp : logcontract(sol);
sol_exp : exp(lhs(sol_simp)) = exp(rhs(sol_simp));
sol_y : solve(sol_exp, y);

d

dx
y=(1− y) y (eq)

log (y)− log (y− 1)=x+%c (sol)

log

(
y

y− 1

)
=x+%c (sol_ simp)

y

y− 1
=%ex+%c (sol_ exp)

[
y=

%ex+%c

%ex+%c− 1

]
(sol_ y)

This result can be transformed using properties of the exponential function by replacing %ex+%c with K%ex,
where K is an arbitrary constant. Maxima must be guided manually to obtain this result, for example with the
command subst(K*%e^x,%e^(x+%c),sol_y);

(% i10) subst(K*%eˆ x,%eˆ (x+%c),sol_y);[
y=

%exK

%exK − 1

]
(% o10)

7.3 Separable differential equation: y′ = f(y)g(x)

Two methods can be used for this type of differential equation:

1. Method 1: solve the equation with ode2, then express y as a function of x whenever possible.

2. Method 2: separate the variables (y on one side and x on the other), integrate each side, then try to
express y as a function of x. In this case, Maxima often needs assistance because the standard commands
are not suited to the situation.

When solving this type of differential equation, one often obtains an equality between an expression involving
y and an expression involving x, but it is often impossible to isolate y in a simple way. Numerical methods can
nevertheless be used to study the resulting solution.

Example 1: Solve the differential equation y′ = yx.

eq: ’diff(y,x) = y*x;
sol: ode2(eq, y, x);
Comment: The command ode2 is used here.

Maxima command:

September 21, 2026 Version 3.0 15/33

https://net124.reltub.ca/yamwi/index.php?c=viotyilhqu0kpsflfepuq&n=2&mode=4&max=ZXEgOiAnZGlmZih5LHgpID0geSooMS15KTsKc29sIDogb2RlMihlcSwgeSwgeCk7CnNvbF9zaW1wIDogbG9nY29udHJhY3Qoc29sKTsKc29sX2V4cCAgOiBleHAobGhzKHNvbF9zaW1wKSkgPSBleHAocmhzKHNvbF9zaW1wKSk7CnNvbF95IDogc29sdmUoc29sX2V4cCwgeSk7CnN1YnN0KEsqJWVeeCwlZV4oeCslYyksc29sX3kpOw--


Result: Click here for Live Demo

(% i2) eq: ’diff(y,x) = y*x;
sol: ode2(eq, y, x);

d

dx
y=x y (eq)

y=%e
x2

2 %c (sol)

eq/y;
sol1:integrate(1/y,y)=integrate(x,x)+K;
solve(sol1,y);
Comment: The equation is solved by separating the variables. The command eq/y: identifies the left-
and right-hand sides to be integrated. The same result is obtained, with a different notation for the
integration constant.

Maxima command:

Result: Click here for Live Demo

(% i3) eq/y;

d
dxy

y
=x (% o3)

(% i4) sol1:integrate(1/y,y)=integrate(x,x)+K;

log (y)=
x2

2
+K (sol1)

(% i5) solve(sol1,y); [
y=%e

x2

2
+K

]
(% o5)

Example 2: Solve the differential equation y′ =
x

y2 + 1

eq: ’diff(y,x) = x/(1 + y^2);
sol: ode2(eq, y, x);
sol2:solve(sol,y)$sol2[3];
Comment: The result is a cubic equation in y3. The solve command gives three roots, two of which are
complex. The real root, which is the third one obtained, is selected.

Maxima command:

Result: Click here for Live Demo

(% i7) eq: ’diff(y,x) = x/(1 + yˆ 2);
sol: ode2(eq, y, x);

d

dx
y=

x

y2+1
(eq)

y3+3y

3
=

x2

2
+%c (sol)

(% i9) sol2:solve(sol,y)$sol2[3];

y=

(√
9x4+36%cx2+36%c2+16

4
+

6%c+3x2

4

) 1
3

− 1(√
9x4 +36%cx2 +36%c2 +16

4 + 6%c+3x2

4

) 1
3

(% o9)

eq*(y^2+1);
sol3:integrate(y^2+1,y)=integrate(x,x)+K;
Comment: To separate the variables, transform the differential equation and integrate each side manually.

Maxima command:

Result: Click here for Live Demo

(% i10) eq*(yˆ 2+1); (
y2+1

) ( d

dx
y

)
=x (% o10)

(% i11) sol3:integrate(yˆ 2+1,y)=integrate(x,x)+K;

y3

3
+ y=

x2

2
+K (sol3)

7.4 Linear differential equation y′ = f(x)y + g(x)

Solve the differential equation y′ = xy + x− 1

eq: ’diff(y,x) = x*y +x-1;
ode2(eq,y,x);
Comment: The solution is expressed in terms of the error function, which is mathematically defined by

erf(z) =
2√
π

∫ z

0
e−t2 dt

Maxima command:

Result: Click here for Live Demo

(% i19) eq: ’diff(y,x) = x*y +x-1;
ode2(eq,y,x);

d

dx
y=xy+x− 1 (eq)

y=%e
x2

2

−

√
π erf

(
x√
2

)
√
2

+%c−%e
−
(

x2

2

) (% o19)

8 Direction field of a first-order ODE

For a differential equation y′ = f(x, y), the derivative y′ represents, at the point (x, y), the slope of the tangent
to the solution curve passing through that point. At each point (x, y) in the plane, a short segment of slope
f(x, y) is drawn. These segments indicate the direction that any solution passing through that point must
follow. The graphical representation of these segments is the direction field of the ODE. Maxima has a built-in
command to plot this direction field:
load(drawdf)$ : load the dedicated package
wxdrawdf(f(x,y),[x,-10,10],[y,-10,10]); plots the field between −10 and 10 on each axis.

load(drawdf)$
wxdrawdf(x - y, [x,-3,3], [y,-3,3])$
Comment: Representation of the direction field of y′ = x− y

Maxima command:

Result: Click here for Live Demo

A particular solution can easily be added to this direction field using the option
[trajectory_at,a,b], which plots the solution satisfying y(a) = b.

load(drawdf)$
wxdrawdf(x-y,[x,-3,3],[y,-3,3],[trajectory_at,-1,3])$

Maxima command:

Result: Click here for Live Demo

9 Numerical approximation of a particular solution of an ODE

9.1 Euler’s method for a first-order ODE

Principle: for a first-order differential equation y′ = f(x, y), a numerical approximation of the solution defined
by the initial condition (x0, y0) can be obtained.

y′0 = f(x0, y0) is the slope of this particular solution at the point with abscissa x0. Knowing the starting point
and the slope, the curve can therefore be approximated by a straight-line segment on the interval [x0, x0+∆x],
where ∆x is a small value. This gives

x1 = x0 +∆x y1 = y0 + y′0 ×∆x

This procedure is then repeated by replacing (x0, y0) with (x1, y1), and so on. An algorithm is thus constructed
that yields the value of y at a given point c. In practice, the interval [x0, c] is divided into n subintervals of
width ∆x. The accuracy of the approximation is directly related to the chosen number n and increases with it.

9.2 Approximating a value of the particular solution

Consider the particular solution of the differential equation y′ = x − y satisfying y(0) = 1. Let us find an

approximation of y(1) using a step size of
1

100
.

/* Euler’s method for y’ = x - y, y(0) = 1 */
/* Parameters */
x0: 0$
y0: 1$
h: 1/100$
nsteps: 100$
/* Function f(x,y) = x - y */
f(x, y) := x - y$
/* Initialization */
x: x0$
y: y0$
/* Euler loop */
for i: 1 thru nsteps do (

y: y + h * f(x, y),
x: x + h

)$
/* Result: approximation of y(1) */
print("Approximation de y(1) par Euler :", float(y))$

Maxima command:

Result: Click here for Live Demo
"Approximation de y(1) par Euler :"

0.732064682546459

kill(x, y)$
eq: ’diff(y,x) = x - y;
sol_exacte: ode2(eq, y, x);
sol_ci: ic1(sol_exacte, x=0, y=1);
y_exact: rhs(solve(sol_ci, y)[1]);
print("Valeur exacte y(1) :", ev(y_exact, x=1, numer))$
Comment: This is compared with the exact solution. To do so, solve the equation with ode2, find the
particular solution with ic1, and compute the value of this function at x = 1.

Maxima command:

Result: Click here for Live Demo

(% i135) /* Comparison with the exact solution */
kill(x, y)$
eq: ’diff(y,x) = x - y;
sol_exacte: ode2(eq, y, x);
sol_ci: ic1(sol_exacte, x=0, y=1);
y_exact: rhs(solve(sol_ci, y)[1]);
print("Exact value y(1) :", ev(y_exact, x=1, numer))$

d

dx
y=x− y (eq)

y=%e−x (%ex (x− 1)+%c) (sol_ exacte)

y=%e−x (%exx−%ex+2) (sol_ ci)

%e−x (%exx−%ex+2) (y_ exact)

"Exact value y(1) :"
0.7357588823428847

9.3 Plotting a curve approximating a particular solution

Principle: to plot, on a given interval, a curve approximating a particular solution of a first-order ODE by
Euler’s method, the interval is subdivided into n parts, and an approximate value of the particular solution y is
computed at each point of these subintervals using the previous method. It then remains to place these points
on a graph and join them.

Plot a curve approximating the particular solution on [0, 2π] of the differential equation

y′ = cos(x) + y with y(0) = 1

The interval is divided into 100 subintervals.

x0: 0.0$y0: 1.0$x_final: 2.0*%pi$nsteps: 100$
h: float((x_final - x0) / nsteps)$
f(x, y) := float(cos(x) + y)$
x: x0$y: y0$
/* Lists to store the points */
X_list: [x]$
Y_list: [y]$
/* Euler loop */
for i: 1 thru nsteps do (

y: y + h * f(x, y), x: x + h,
X_list: endcons(x, X_list),
Y_list: endcons(y, Y_list)

)$
wxplot2d([discrete, X_list, Y_list],

[style, [lines, 2, red]],
[xlabel, "x"],
[ylabel, "y"],
[title, "Euler approximation for y’ = cos(x) + y, y(0) = 1"],
[legend, "Curve approximating the solution"]

Maxima command:

)$
Comment: To reduce Maxima’s computational load, approximate values of the computed points are used
via the float command, because exact arithmetic is very slow.

Result: Click here for Live Demo

9.4 Approximating a solution with the built-in rk command

For a first-order ODE, the built-in Maxima command rk applies the classical fourth-order Runge–Kutta method
to compute successive numerical approximations of the solution on a given interval. This interval is divided
into n subintervals, on each of which four slope calculations are performed. This method is therefore more
accurate than Euler’s method presented previously. The syntax of this command is:
rk(equation, function, initial_value, [variable, start, end, step]);

Find an approximation of the solution of y′ = x− y with y(0) = 1 on [0, 1] with a step size of 0.01.

rk( x - y, y, 1,[x, 0, 1, 0.4]);
listepoints:rk( x - y, y, 1,[x, 0, 1, 0.01])$
wxdraw2d(

title = "Approximation de la solution avec rk ",
xlabel = "x",
ylabel = "y",
color = blue,
point_type = filled_circle,
point_size = 0.3,
line_width = 2,
points_joined = true,
points(listepoints)

)$
Comment: rk returns a list of points. The first command illustrates the list obtained when the step size is
0.25. The list of points is then generated for the desired subdivision, and the points are plotted and joined
to obtain the curve approximating the particular solution of the differential equation.

Maxima command:

Result: Click here for Live Demo

(% i19) rk( x - y, y, 1,[x, 0, 1, 0.4]);

[[0.0 , 1.0] , [0.4 , 0.7408] , [0.8 , 0.69887232] , [1.0 , 0.7359367307946667]] (% o19)

9.5 Approximating a solution with the rkf45 command

For a first-order ODE, the rkf45 command implements the fourth- and fifth-order Runge–Kutta–Fehlberg
method with an adaptive step size. This simply means that the method adapts the subdivision of the interval
on which an approximate value of the solution is sought, making it finer by dynamically adjusting the step size
according to the estimated error (we recall that the rk command divides the interval into subintervals of equal
width). The result is therefore more accurate, and the computation is faster. The syntax is:
load(rkf45)$: loads the package for this command
rkf45(expr, var, init, [t, t0, tf], options)
List of the main possible options:

• report: if true, displays a report on the solution process

• absolute_tolerance: absolute error tolerance

• max_iterations: maximum number of iterations

Example 1: Solving y′ = x(y − 1) + 3 with y(0) = −2

load(rkf45)$
sol : rkf45(x*(y-1)+3, y, -2, [x, 0, 4],

report=true,
absolute_tolerance=1e-8)$

makelist(sol[i],i,1,4);
Comment: We chose the options to display the report for the solution of this ODE and to set the error
to 10−8. The command returns the list of points approximating the solution, which can be used for a
graphical plot. Here, only the first 4 points of this list are displayed using the makelist command.

Maxima command:

Result: Click here for Live Demo

(% i22) load(rkf45)$
sol : rkf45(x*(y-1)+3, y, -2, [x, 0, 4],
report=true,
absolute_tolerance=1e-8)$
makelist(sol[i],i,1,4);

"——————————————————"

"Info: rkf45:"

" Integration points selected:"
709

" Total number of iterations:"
713

" Bad steps corrected:"
5

" Minimum estimated error:"
3.07524023430592910−10

" Maximum estimated error:"
8.62248960483550810−9

"Minimum integration step taken:"
8.83091253070400510−4

"Maximum integration step taken:"
0.09864741067221602

"——————————————————"

[ [0.0 ,− 2.0] , [0.04 ,− 1.8823369398295748] , [0.0964347253301173 ,− 1.723779310357251] (% o22)

, [0.15751564184437233 ,− 1.560974036056569] ]

Example 2: Differential equation for the pendulum θ′′(t) + sin(θ(t)) = 0.
To use rkf45, this second-order equation is transformed into a system of two first-order equations by setting

ω(t) = θ′(t), which gives

{
θ′(t) = ω(t),

ω′(t) = − sin
(
θ(t)

)
.

We set the initial values θ(0) = 0.1 and ω(0) = 0, which corresponds to the classical motion of a pendulum.

load(rkf45)$
sol2:rkf45([omega, -sin(theta)], [theta, omega],

[0.1, 0], [t, 0, 20],
absolute_tolerance=1e-6,
report=true)$

nbpoints:length(sol2)$
listepoints:makelist(sol2[i],i,1,80)$
wxdraw2d(

title = "Approximation of the solution with rkf45 ",
xlabel = "x",

Maxima command:

ylabel = "y",
color = blue,
point_type = filled_circle,
point_size = 0.3,
line_width = 2,
points_joined = false,
points(listepoints)

)$
Comment: The rkf45 command generates the solution points, which are converted into a list with
makelist so that a graph approximating the solution can be plotted.

Result: Click here for Live Demo

(% i28) load(rkf45)$
sol2 : rkf45([omega, -sin(theta)], [theta, omega],
[0.1, 0], [t, 0, 20],
absolute_tolerance=1e-6,
report=true)$
nbpoints:length(sol2)$
listepoints:makelist(sol2[i],i,1,80)$
wxdraw2d(
title = "Approximation of the solution with rkf45 ",
xlabel = "x",
ylabel = "y",
color = blue,
point_type = filled_circle,
point_size = 0.3,
line_width = 2,
points_joined = false,
points(listepoints)
)$

"——————————————————"

"Info: rkf45:"

" Integration points selected:"
80

" Total number of iterations:"
79

" Bad steps corrected:"
0

" Minimum estimated error:"
1.746633458032021310−12

" Maximum estimated error:"
6.08494713437850110−7

September 21, 2026 Version 3.0 24/33

https://net124.reltub.ca/yamwi/index.php?c=orymnwnwsvreuoddu4cxh&n=2&mode=2&max=--


"Minimum integration step taken:"
0.011093833762060967

"Maximum integration step taken:"
0.27222421225747034

"——————————————————"

10 The built-in desolve command

There is another native Maxima command for solving a differential equation. This is desolve, which uses
the Laplace transform for this purpose. In particular, this command can handle systems of linear differential
equations and problems with initial conditions given at t = 0. In the syntax of this command, the unknown
functions must be written in functional form (y(x) instead of y).
desolve([eq1, eq2, ...], [y(x), z(x), ...]);

Example 1: Solve y′ + y = e−x with y(0) = 2.

de: ’diff(y(x), x) + y(x) = exp(-x);
sol: desolve(de, y(x));
sol_ci: ev(sol, y(0) = 2);
Comment: The general solution is expressed in terms of the initial value at 0. If the particular solution is
defined by this initial value at 0, it can be obtained directly by replacing y(0) with its value. Otherwise,
y(0) is treated as a constant to be determined from the given condition, as in Chapter 5.

Maxima command:

Result: Click here for Live Demo

(% i11) de: ’diff(y(x), x) + y(x) = exp(-x);
sol: desolve(de, y(x));
sol_ci: ev(sol, y(0) = 2);

d

dx
y(x)+ y(x)=%e−x (de)

y(x)=%e−xx+%e−x y(0) (sol)

y(x)=%e−xx+2%e−x (sol_ ci)

Example 2: Solve y′′ − y = ex with y(0) = 1 and y′(0) = 0.
Here, the atvalue command is used to define the initial values that will be used by desolve to find the
particular solution. These commands atvalue must be defined before executing desolve.
atvalue(y(x), x=0, 1); and atvalue(’diff(y(x),x), x=0, 0);

atvalue(y(x), x = 0, 1)$
atvalue(’diff(y(x), x), x = 0, 0)$
de2: ’diff(y(x), x, 2) - y(x) = exp(x);
sol2: desolve(de2, y(x));
Comment: The particular solution is obtained directly.

Maxima command:

Result: Click here for Live Demo

(% i50) atvalue(y(x), x = 0, 1)$
atvalue(’diff(y(x), x), x = 0, 0)$
de2: ’diff(y(x), x, 2) - y(x) = exp(x);
sol2: desolve(de2, y(x));

d2

dx2
y(x)− y(x)=%ex (de2)

y(x)=
%exx

2
+

%ex

4
+

3%e−x

4
(sol2)

Example 3:
Solve the system of differential equations: 

dx

dt
= x+ y

dy

dt
= 4x− 2y

with initial conditions x(0) = 1 and y(0) = 0.

atvalue(x(t), t=0, 1)$
atvalue(y(t), t=0, 0)$
eq1: ’diff(x(t),t) = x(t) + y(t);
eq2: ’diff(y(t),t) = 4*x(t) - 2*y(t);
sol3: desolve([eq1,eq2], [x(t),y(t)]);
Comment: in this example, x and y denote functions, while the independent variable is t.

Maxima command:

Result: Click here for Live Demo

(% i79) atvalue(x(t), t = 0, 1)$
atvalue(y(t), t = 0, 0)$
eq1: ’diff(x(t), t) = x(t) + y(t);
eq2: ’diff(y(t), t) = 4*x(t) - 2*y(t);
sol3: desolve([eq1, eq2], [x(t), y(t)]);

d

dt
x(t)= y(t)+x(t) (eq1)

d

dt
y(t)= 4x(t)− 2 y(t) (eq2)

[
x(t)=

4%e2t

5
+

%e−(3t)

5
, y(t)=

4%e2t

5
− 4%e−(3t)

5

]
(sol3)

11 Going further: the contrib_ode package

This package extends the capabilities of ode2 for certain ODEs that ode2 cannot solve, in particular some
nonlinear first-order equations and some homogeneous linear second-order equations with variable coefficients,
which may have several families of solutions. For first-order ODEs, contrib_ode first calls ode2, then tries
other methods (factorization, Clairaut equations, Lagrange equations, Riccati equations, Abel equations, . . . ).
The essential commands of this package are:
load(contrib_ode); load the package
contrib_ode(equation, function, variable); attempts to solve the differential equation
ode_check(equation, candidate_solution); tests whether the function “candidate_solution”
satisfies the differential equation or not.

Example 1: Solve xy′′ − (1 + xy)y′ + y = 0

load(contrib_ode)$
eq: x*’diff(y,x)^2 - (1+x*y)*’diff(y,x) + y = 0;
ode2(eq,y,x);
contrib_ode(eq,y,x);
ode_check(eq,[y=log(x)]);
Comment: First try ode2, which fails to solve the differential equation, then use contrib_ode for
comparison. The last command tests whether the function log is indeed a solution; this is the case if the
result is 0.

Maxima command:

Result: Click here for Live Demo

(% i29) eq: x*’diff(y,x)ˆ 2 - (1 + x*y)*’diff(y,x) + y = 0;
ode2(eq,y,x);
contrib_ode(eq,y,x);
ode_check(eq,[y=log(x)]);

x

(
d

dx
y

)2

− (xy+1)

(
d

dx
y

)
+ y=0 (eq)

x

(
d

dx
y

)2

− (xy+1)

(
d

dx
y

)
+ y=0 (% t27)

"first order equation not linear in y’"

false (% o27)

x

(
d

dx
y

)2

− (xy+1)

(
d

dx
y

)
+ y=0 (% t28)

"first order equation not linear in y’"

[y= log (x)+%c , y=%ex%c] (% o28)

0 (% o29)

Example 2: Solve y′ = y2 − 2xy + x2 + 1

eq2: ’diff(y,x) = y^2 - 2*x*y + x^2 + 1;
contrib_ode(eq2, y, x);
Comment: This is a Riccati equation.

Maxima command:

Result: Click here for Live Demo

(% i34) eq2: ’diff(y,x) = yˆ 2 - 2*x*y + xˆ 2 + 1;
contrib_ode(eq2, y, x);

d

dx
y= y2− 2xy+x2+1 (eq2)

[[
x=

1√
%t− 1

+%c , y=x−
√

%t− 1

]
,

[
x=%c− 1√

%t− 1
, y=x+

√
%t− 1

]]
(% o34)

This package also includes the command odelin, which solves certain homogeneous linear second-order
equations with variable coefficients. The syntax is:
odelin(equation, function, variable);

Example 3: Solve (x2 + 1)y′′ + xy′ − y = 0

eq3: (x^2 + 1)*’diff(y,x,2) + x*’diff(y,x) - y = 0;
sol3: odelin(eq3, y, x);
Comment: odelin returns a list of functions forming a basis of the solution space. The general solution
is therefore generated by a linear combination of the functions returned by odelin.

Maxima command:

Result: Click here for Live Demo

(% i9) eq3: (xˆ 2 + 1)*’diff(y,x,2) + x*’diff(y,x) - y = 0;
sol3: odelin(eq3, y, x);

(
x2+1

) ( d2

dx2
y

)
+x

(
d

dx
y

)
− y=0 (eq3)

{
x ,
√

x2+1
}

(sol3)

sol4:listify(sol3);
y=%k1*sol4[1]+%k2*sol4[2];
Comment: The result, which is a set, is transformed into a list using the listify command, then the
general solution of the differential equation is generated.

Maxima command:

Result: Click here for Live Demo

(% i13) sol4:listify(sol3); [
x ,
√

x2+1
]

(sol4)

(% i14) y=%k1*sol4[1]+%k2*sol4[2];

y=%k2
√
x2+1+%k1x (% o14)

Remarks on the results returned by contrib_ode
Unlike ode2, the contrib_ode command may return a list containing several solutions or several families
of solutions. The different elements of the list returned by Maxima must therefore be examined.
The ode_check command verifies a proposed solution by substituting it into the differential equation. It then
returns the residual obtained after simplification. A result equal to 0 indicates that the solution satisfies the
differential equation within the simplifications performed by Maxima:

ode_check(equation, solution);

This verification does not remove the need to study the conditions under which the solution is valid: its domain
of definition, excluded values, intervals of validity, and any initial conditions.
For some nonlinear equations, contrib_ode may provide a parametric solution or use an auxiliary variable
such as %t. This variable must not be confused with the independent variable of the differential equation.
The variable method may also provide information about the method used by contrib_ode:

12 The odes package by Aleksas Domarkas

The odes package complements Maxima’s native commands for linear differential equations with constant
coefficients. Its main interest is that it makes it possible to handle equations of order greater than two, as well
as to impose initial conditions for these equations. It does not replace ode2, desolve, or contrib_ode,
which address other classes of problems. The package is loaded with the command:
load(odes);
Once the package has been loaded, several new commands for solving differential equations become available.
These include, for example:

• odeL: solves linear ODEs with constant coefficients;

• odeL_ic: solves linear ODEs with constant coefficients and initial conditions;

• fs: finds a fundamental system of solutions for the associated homogeneous equation;

Example 1: Solve the third-order differential equation:

d3y

dx3
+ 2

d2y

dx2
− dy

dx
= 0

load(odes)$
eq:’diff(y,x,3) - 2*’diff(y,x,2) + ’diff(y,x) = 0;
odeL(eq, y, x);
Comment: Homogeneous third-order equation.

Maxima command:

Result: Click here for Live Demo

(% i5) eq : ’diff(y,x,3) - 2*’diff(y,x,2) + ’diff(y,x) = 0;
odeL(eq, y, x);

d3

dx3
y− 2

(
d2

dx2
y

)
+

d

dx
y=0 (eq)

y=%exC3x+%exC2+C1 (% o5)

Example 2: Solve
d4y

dx4
− 5

d2y

dx2
+ 4y = 0.

eq2:’diff(y,x,4) - 5*’diff(y,x,2) + 4*y = 0;
odeL(eq2, y, x);
Comment: Homogeneous fourth-order equation.

Maxima command:

Result: Click here for Live Demo

(% i9) eq2 : ’diff(y,x,4) - 5*’diff(y,x,2) + 4*y = 0;
odeL(eq2, y, x);

d4

dx4
y− 5

(
d2

dx2
y

)
+4y=0 (eq2)

y=%e2xC4+%exC3+%e−xC2+%e−(2x)C1 (% o9)

Example 3: Solve
d4y

dx4
+ 8

d2y

dx2
+ 16y = sin(x)3 + x7.

eq3 : ’diff(y,x,4) + 8*’diff(y,x,2) + 16*y= x^7 + sin(x)^3;
sol3 : odeL(eq3, y, x);
Comment: Non homogeneous equation with constant coefficients.

Maxima command:

Result: Click here for Live Demo

(% i14) eq3 : ’diff(y,x,4) + 8*’diff(y,x,2) + 16*y= xˆ 7 + sin(x)ˆ 3;
sol3 : odeL(eq3, y, x);

d4

dx4
y+8

(
d2

dx2
y

)
+16y=sin (x)3+x7 (eq3)

y=−
(
24 sin (3x)− 200 sin (x)− 150x7+3150x5− 23625x3+47250x

2400

)
(sol3)

+C4x sin (2x)+C3 sin (2x)+C2x cos (2x)+C1 cos (2x)

Example 4: Solve
d4y

dx4
− d3y

dx3
− d2y

dx2
+

dy

dx
= 0

with y(0) = 2, y′(0) = 1, y′′(0) = 0, and y(3)(0) = 3.

eq4: ’diff(y,x,4) - ’diff(y,x,3)- ’diff(y,x,2) + ’diff(y,x) = 0;
sol4 : odeL_ic(eq4, y, x, [0, 2, 1, 0, 3]);
define(f(x),rhs(sol4));
[f(0),subst(0,x,diff(f(x),x,1)),
subst(0,x,diff(f(x),x,2)),subst(0,x,diff(f(x),x,3))];
Comment: We use the odeL_ic command, giving in the argument list the initial value of x, followed
by the values of y(0), y′(0), y′′(0), and y(3)(0). We then check the values of the successive derivatives at
0 for the particular solution obtained.

Maxima command:

Result: Click here for Live Demo

(% i143) eq4: ’diff(y,x,4) - ’diff(y,x,3)- ’diff(y,x,2) + ’diff(y,x) = 0;
sol4 : odeL_ic(eq4, y, x, [0, 2, 1, 0, 3]);
define(f(x),rhs(sol4));
[f(0),subst(0,x,diff(f(x),x,1)),
subst(0,x,diff(f(x),x,2)),subst(0,x,diff(f(x),x,3))];

d4

dx4
y− d3

dx3
y− d2

dx2
y+

d

dx
y=0 (eq4)

y=%exx−%ex−%e−x+4 (sol4)

f(x) :=%exx−%ex−%e−x+4 (% o142)

[2 , 1 , 0 , 3] (% o143)

Example 5: Solve
d4y

dx4
+ 2

d3y

dx3
+ 2

d2y

dx2
+ 2

dy

dx
+ y = 0.

eq5:’diff(y,x,4)+2*’diff(y,x,3)+2*’diff(y,x,2)+2*’diff(y,x)+y=0;
fs(eq5, y, x);
base5 : listify(fs(eq5, y, x))$
sol5 : %k1*base5[1] + %k2*base5[2]+ %k3*base5[3] + %k4*base5[4];
Comment: We use the fs command to generate a basis of solutions, and then construct the set of all
solutions as a linear combination of the elements of this basis.

Maxima command:

Result: Click here for Live Demo

(% i153) eq5 : ’diff(y,x,4) + 2*’diff(y,x,3) + 2*’diff(y,x,2) + 2*’diff(y,x) + y = 0;
fs(eq5, y, x);
base5 : listify(fs(eq5, y, x))$
sol5 : %k1*base5[1] + %k2*base5[2]+ %k3*base5[3] + %k4*base5[4];

d4

dx4
y+2

(
d3

dx3
y

)
+2

(
d2

dx2
y

)
+2

(
d

dx
y

)
+ y=0 (eq5)

[
%e−x ,%e−xx , cos (x) , sin (x)

]
(% o151)

%k4 sin (x)+%k3 cos (x)+%e−x%k2x+%e−x%k1 (sol5)

For any remarks or comments on this document, you can send an email to michel.gosse@free.fr.

logcontract, 15

ODE solver, 6

Particular solution of an ODE, 9

